Threshold Calculator Description

This is a description of the key variables in the Threshold Calculator worksheet on the Thousand Home Challenge spreadsheet, version 1.2, and how they were calculated.

Each variable is identified by its cell row and column identifier on the worksheet in which it was calculated (e. g., C28). A list of other cells used to calculate the variable is given, followed by a brief description of the logic and the Excel formula itself (sometimes simple, sometimes quite involved).

In what follows, input variables from the User Interface are in **bold type**, and variables calculated elsewhere in the threshold calculator are in *italics*, with the actual cell name in parentheses. Output variables (the bottom line returned to the User Interface worksheet) are in *CAPITALIZED ITALICS*. (Italicized because they are calculated by the spreadsheet, and capitalized because they are returned to the User Interface.)

In some cases for the sake of clarifying the logic, I have created variables (X, Y, etc.) that do not actually exist in the Excel spreadsheet, to create separate logical steps to help understand and digest a long Excel statement, although in fact Excel treats each cell as a single (albeit sometimes complex) statement.

All cell names refer to cells in Threshold Calculator worksheet, unless specified otherwise.

(Note for those reading the Excel formulas: several constants, such as F6, F8, and F9 are used solely for the purpose of converting from one set of units to another. In the U. S. context we are considering X square feet, degrees Fahrenheit, and kilowatt hours X these constants can be disregarded, they are always set to 1, and they don't affect the basic logic anyway, only the conversion of units.)

Weather Station (C11)

Input:
Matched Station (User interface, F7)
Override-Station (User interface, F8)

Weather station is either the **Matched Station** or the **Override-station**. If the **Override-station** is "use zip code,' then the **Matched Station** is used; otherwise, the **Override-station** is used.

(The user enters the **Home's 5-digit zip code** which determines the **Matched Station** based on a lookup to the "zip2station" table (contained in the "Weather data, zip codes" worksheet). The zip code match is based on proximity. However, the user can, in the User Interface, override whatever value is calculated based on their zip code with a weather station of their own choosing, namely, the **Override-station**.)

Heating Degree Days (C13)
Cooling Degree Hours (C14)
Dewpoint Degree Hours (C15)

All three cells referring to climate data are determined by a separate table lookup using the *Weather Station* (C11).

The Excel formulas are respectively:
=$F$9*VLOOKUP(C$11,weather,2,FALSE)
=$F$9*VLOOKUP(C$11,weather,4,FALSE)
=$F$9*VLOOKUP(C$11,weather,3,FALSE)

Weather refers to a data table on the Weather data, zip codes worksheet. F9 is used to determine whether Fahrenheit or Centigrade degrees should be used.

Number of households (C16)
Number of occupants (C17)
Finished floor area (C18)
Multifamily attached % (C19)

These four variables are set equal to the corresponding values on the User Interface:

Number of households in building (User Interface, C10)
Number of occupants (User Interface, C9)
Finished floor area (ft2) (User Interface, C8)
Attached home, % common (User Interface, C11)

There's no decision logic here, the values are just set equal to the values entered from the User Interface.

Shell Area Assumption (C21)

Input:
Finished Floor Area (C18)
The intent of this variable is to express the area exposed to the outside, namely five faces of the house: the roof and the four walls, based on the *Finished floor area* (C18).

If the *Finished Floor Area* is less than 1400 (square feet):
Calculate the *Shell Area Assumption* =
(35 * Square root of the *Finished Floor Area*) + *Finished Floor Area*
Otherwise,
Calculate the *Shell Area Assumption* =
(70 * Square root of the (*Finished Floor Area* / 2)) + (*Finished Floor Area* / 2)

The Excel formula is:
=$F$6*IF((C18/$F$6)<1400,(35*SQRT(C18/$F$6)+(C18/$F$6)),2*35*SQRT((C18/$F$6)/2)+(C18/$F$6)/2)

Rationale: the area covered by the roof is equal to the finished floor area. Add to that the area covered by four walls as follows: a house with a perfectly square floor plan and walls 8 feet high would have a wall area of "4 * 8 * square root of the finished floor area" = "32 * square root of the finished floor area." A typical rectangular shape for a house (different aspect ratio) would increase the total wall area by about 10%, so we use a figure of 35 rather than 32. That gives us the shell area assumption, the total area exposed to the outside world.

If the house has two stories, you need to double the area allocated to the height (2 * 35 = 70), and cut the finished floor area dimension in half when calculating the roof area and the wall area. There is a slight built-in efficiency advantage to having the house on two stories instead of one story (the shell area assumption for two stories is about 90% of the same assumption for one story), so the allowance is slightly less for two stories.

The assumption is that houses with more than 1400 square feet are likely to have two stories. There is no good way of making the calculation more accurate (or have a special adjustment for 3 or more stories) without asking for additional input and making the spreadsheet more complex.

F6 is a constant used to switch between different units of measurement.

Heating if electric (C24)

This figure is the heating allowance for the house if electricity is used for heat. It is also a key input in calculating *Heating if fossil* (D24) (see below).

Inputs:
Shell Area Assumption (C21)
Heating degree days (C13)
Multifamily/attached percentage (C19)
If *Heating degree days* are less than 1500, then *Heating if electric* is 0.

Otherwise calculate *Heating if Electric* =
(*Shell Area Assumption*
* (1 - (*Multifamily/attached percentage* * 0.75))
* 0.028
* (0.75 * *Heating degree days* * 24))
/ 3412

"0.028" is the Heating Shell average U-value. This is the BTUs / square feet / hour / degree F.
"24" is the hours in a day.
"3412" is a conversion ratio; 1 kilowatt hour = 3412 BTUs. Dividing the number of BTUs by 3412 gives the number of kilowatt hours.
"0.75" is an adjustment factor to reflect the fact that buildings operate at a lower balance point temperature than 65 degrees F. The "balance point temperature" is the outdoor temperature at which building heat gains (solar, appliance waste heat) are dissipated at a rate that creates a desired indoor air temperature.

The Excel formula is:
=$F$8*C$21/$F$6*(1-C$19*0.75)*IF(C$13/$F$9<1500,0,0.028)*(0.75*(C13/$F$9)*24)/3412

F6, F8, and F9 are used for unit conversions, they are all 1 if you are dealing with square feet, kilowatt hours, and Fahrenheit degrees.

Cooling if electric (C25)

Inputs:
Shell Area Assumption (C21)
Multifamily / Attached % (C19)

Dewpoint degree hours (C15)
Cooling degree hours (C14)

If the *Dewpoint degree hours* (C15) + (*Cooling degree hours* (C14) / 2) < 3000, then the cooling allowance is 0 (no cooling required).

Otherwise, the *Cooling if electric allowance* =
(*Shell Area Allowance*
* (1 - *(Multifamily/Attached %* * 0.75))
* 0.055
* ((1/2 * *Cooling Degree Hours*) + *Dewpoint Degree Hours*) X but not more than 50,000, if this value is more than 50,000, use 50,000 instead)
/ 3412

"3412" is a conversion ratio; 1 kilowatt hour = 3412 BTUs. Dividing the number of BTUs by 3412 gives the number of kilowatt hours.

"0.055" is the cooling pseudo U value in BTUs / square feet / hour / degree F.

"50,000" is the maximum cooling climate severity so that tropical climates don't wind up with an extremely high cooling allowance.

The Excel formula is:
=$F$8*C$21/$F$6*(1-C$19*0.75)*IF((C$15+0.5*C$14)/$F$9<3000,0,0.055)*(MIN(50000,0.5*C$14+C$15))/$F$9/3412

Note that F6, F8, and F9 are unit converters, always 1 in this context.

Hot Water if electric Allowance (C26)

Compute *Hot Water if electric allowance* based on:
Number of occupants (C17)
Heating degree days (C13)
Cooling degree hours (C14)

1. Compute X =
(the hot water allowance for persons 1-2, namely 10) * 1 (or 2, if there are at least two people))
+ (the hot water allowance for additional persons, namely 7) * the number of people over 2)

2. Compute Y =
(6 + 70.6 + (0.0001732* *Cooling degree hours* (C14)))
- (0.00223 * *Heating Degree Days* (C13))
- (0.104 * square root of *Heating Degree Days*) .
If Y is < 45, set Y = 45.

3. Compute the *Hot Water if electric allowance* =
(X * 365 * 8.3 * (125 - Y)) / 3412

“365” is the number of days in the year to convert a daily usage to an annual usage. “8.3” is the specific heat of 1 gallon of water.
“125” is the assumed heat of hot water in degrees F.
“3412” is a conversion ratio; 1 kilowatt hour = 3412 BTUs. Dividing the number of BTUs by 3412 gives the number of kilowatt hours.
“45” is the minimum average cold water inlet temperature in degrees F.
The variable “Y” and the associated constants are based on various standard cold water inlet models and statistical relationships for these climate parameters. Its purpose is to calculate the average cold water inlet temperature in degrees F, based on the climate (the heating degree days and cooling degree hours).

NOTE: the variables X and Y do not exist in the Excel formula. They are added for clarity to break the computation down into several steps.

The Excel formula is:
=$F$8*(MIN(2,C$17)*10+MAX(0,C$17-2)*7)*365*8.3*(125-MAX(45,(6+70.6+0.0001732*(C14/$F$9)-0.00223*(C13/$F$9)-0.104*SQRT(C13/$F$9))))/3412

F8 and F9 are used for unit conversions and are always 1 in this context.

All Else if electric (C27)

This is based on:
Number of households (C16)
Number of occupants (C17)
Finished Floor Area (C18)

This is calculated as follows:
(*Number of households* * Base per home, namely 400)
+ ((Additional allowance per square foot, namely 0.2) * *Finished Floor Area*)
+ ((Additional allowance per person 1 and 2, namely 500)
* 1 (or * 2 if *number of occupants* >= 2))
+ (Additional allowance per person over 2, namely 200 * *Number of occupants* over 2)

The Excel formula is:
=$F$8*(C$16*400+$C$18*0.2+500*MIN(2,C17)+200*MAX(0,C19-2))

F8 is for unit conversion and is always 1 in this context.

MAXIMUM TOTAL FOR ELECTRIC (C28)

Uses these inputs:
Heating if electric (C24)
Cooling if electric (C25)
Hot Water if electric (C26)
All Else if electric (C27)

This total is simply the sum of the inputs. The units are kWh.

The Excel formula is:

=SUM(C24:C27)

Heating if fossil (D24)

This refers to any homes heated with any fossil fuel, e. g. natural gas, oil, or coal, in addition to wood.

This uses these inputs:
Heating if electric (C24)

This multiplies the electric allowance by two. The heating allowance for the fossil /wood heated allowance is twice that of an electrically heated home since the site energy input needed to heat a home with a fossil or wood heat source (based on combustion) is twice as much as can be obtained using compressor-based heat pump technology available for electrically heated homes. Without this adjustment, the THC heating allowance threshold would be biased against homes using fossil / wood heat

This is calculated as follows:
2 * *Heating if electric*

The Excel formula is:
=2*C24

Cooling if fossil (D25)
Hot Water if fossil (D26)
All Else if fossil (D27)

These values are all set equal to, respectively,
Cooling if electric (C25)
Hot Water if electric (C26)
All Else if electric (C27)

The values for cooling, hot water, and all else is the same for homes using electricity for heat and homes using gas, oil, some other fossil fuel, or wood for heat.

The formulas are, respectively,
=C25
=C26
=C27

MAXIMUM TOTAL FOR FOSSIL (D28)

Uses these inputs:
Heating if fossil (D24)
Cooling if fossil (D25)
Hot Water if fossil (D26)
All Else if fossil (D27)

This total is simply the sum of the inputs. The units are kWh.
This total will be the same as the *MAXIMUM TOTAL FOR ELECTRIC* except for the difference between *Heating if fossil* and the *Heating if electric*.

The Excel formula is:
=SUM(D24:D27)

Other variables not integral to the calculation

There are other variables in the Threshold Calculator worksheet which are not integral to the calculation of the performance threshold, but which may provide interesting information. The meaning of these variables is usually self-explanatory. Unit conversion constants are omitted and all units are assumed to be kWh and ft2.

"What if" Table Reference (F14)

The "What if" table reference cell allows the user to see what values the THC calculator would give for the cell specified, but varying the house size and the location (weather station). This reference cell (F14) controls the values in the "what If" table (columns F through I, rows 14 through 24).

For example, you could display the hot water requirements for a house just like that described in the User Interface, but located in 10 different cities, and with 3 different sizes. If you wanted to see the "Heating if Electric" values for these 30 different size/location parameters, you would enter "=C24". Or, to see the "Maximum Total if Fossil/Wood" you would enter "=D28".

The column descriptions for the "what if" table are "house size" (cells G14, H14, and I14) of 1200, 2400, or 3600. The row descriptions are various "weather stations" (cells F15 through F24), ranging from Phoenix, Arizona to Duluth, Minnesota. Users may edit these column and row descriptions (the weather stations and house sizes). The reference cell specifies the cell for which data is being calculated in these various house size and location configurations. The effect is to display, as values in the "what if" table, the contents of what the cell specified in F14 would contain for an identical house, but with the specified size and location (1200 square feet in Phoenix, or 3600 square feet in Duluth).

To use this reference, first enter the relevant data for the house in question in the User Interface (number of occupants, number of households, and so forth). Then, go to the Threshold Calculator worksheet and enter the cell location you want to see data on, preceded by an "equal" sign ("="). The "what if" table would then give you the results which would appear in the cell specified, if the house size (finished floor area) and location (weather station), were varied to the values specified in the row and column.

Other Informational Variables

These variables are not essential to computing the Maximum Total Electric or Maximum Total Fossil/Wood thresholds, but do give interesting information. In two cases (D38 and D39) variables are used to display subtotals in the User Interface.

1. *Total kWh per ft2 – electric* (C30)
= *MAXIMUM TOTAL ELECTRIC* (C28) / *Finished Floor Area* (C18)

2. *Total kWh per ft2 – fossil / wood* (D30)
= *MAXIMUM TOTAL FOSSIL / WOOD* (D28) / *Finished Floor Area* (C18)

3. *Heating kWh/yr per ft2 – electric* (C31)
= *Heating if electric* (C24) / *Finished Floor Area* (C18)

4. *Heating kWh/yr per ft2 – fossil / wood* (D31)
= *Heating if fossil* (D24) / *Finished Floor Area* (C18)

5. *Space conditioning kWh/yr per ft2 – electric* (C32)
= (*Heating if electric* (C24) + *cooling if electric* (C25))
/ *Finished Floor area* (C18)

6. *Space conditioning kWh/yr per ft2 – fossil / wood* (D32)
= (*Heating if fossil* (D24) + *cooling if fossil* (D25)) / *Finished Floor area* (C18)

7. *Total kWh per person – electric* (C33)
= *Heating if electric* (C24) / *Number of occupants* (C17)

8. *Total kWh per person – fossil* (D33)
= *Heating if fossil* (D24) / *Number of occupants* (C17)

9. *EPA Home Energy Yardstick, Avg kWh/yr* (C35)
= ((34.09* *Finished Floor area* (C18))
+ (17965* *Number of occupants* (C17))
+ (9.633* *Heating Degree Days* (C13))
+ (2.48**Cooling Degree hours* (C14)))
/(3.412*3.34)

10. *% of EPA Home Energy Yardstick, Avg Use – electric* (C36)
= *MAXIMUM TOTAL ELECTRIC* (C28)
/ *EPA Home Energy Yardstick Avg* (C35)

11. *% of EPA Home Energy Yardstick, Avg Use – fossil* (D36)
= (((3.34*3.412
* *SUBTOTAL MMBTU/YR FOSSIL/WOOD HEAT/HOT WATER* (D39))
+ (1047 * *SUBTOTAL MMBTU/YR FOSSIL/WOOD HEAT/HOT WATER* (D38))
/ (*EPA Home Energy Yardstick, Avg kWh/yr* (C35) * 3.34 * 3.412)

12. *SUBTOTAL MMBTU/YR FOSSIL/WOOD HEAT/HOT WATER* (D38)
= 0.003412 * ((2* *Heating if electric* (C24)) + *Hot Water if electric* (C26))

13. *SUBTOTAL KWH /YR (EXCLUDING HEAT/HOT WATER)* (D39)
= ((*MAXIMUM TOTAL ELECTRIC* (C28) – *Hot water if electric* (C26))
– (*Heating if electric* (C24))

14. *Total MMBtu yr – electric* (C40)
= *MAXIMUM TOTAL ELECTRIC* (C28) * 0.003412

15. *Total MMBtu yr – fossil* (D40)
= *(SUBTOTAL KWH /YR (EXCLUDING HEAT/HOT WATER)* (D39) *0.003412)
+ *SUBTOTAL MMBTU/YR FOSSIL/WOOD HEAT/HOT WATER* (D38))